

PROGRAMACIÓN DECLARATIVA

PROGRAMACIÓN LÓGICA

Tema PL2: El lenguaje PROLOG, aspectos básicos

2. Sintaxis

Grado en Ingeniería Informática

URJC

Ana Pradera


Contenido

- 1 ELEMENTOS BÁSICOS DE PROLOG
- 2 TÉRMINOS
- 3 PREDICADOS
- 4 PROGRAMAS
 - Ejemplo: programa “Abuelas/os”
 - Ejemplo: programa “Amistades/Enemistades”
 - Ejercicios: programa “Amistades/Enemistades”
 - Ejemplo: programa en lógica “pura” para sumar naturales
 - Ejercicios: programación en lógica “pura”
- 5 CONSULTAS
 - Ejercicios: consultas al programa “Amistades/Enemistades”
- 6 RESUMEN
 - Ejercicios: ejercicios 1 y 2 de la Práctica nº 1

ELEMENTOS BÁSICOS DE PROLOG

- **Términos**: sirven para representar *objetos* y constituyen los argumentos de los predicados.
- **Predicados**: tienen términos como argumentos y describen *propiedades o relaciones* entre sus argumentos.
- **Programas**: están compuestos por **cláusulas de Horn positivas**, fórmulas lógicas que involucran predicados y por lo tanto establecen propiedades o relaciones entre objetos. Actúan como premisas en los razonamientos.
- **Consultas a programas**: son fórmulas lógicas existenciales, denominadas **cláusulas de Horn negativas**, que involucran a uno o más predicados de un programa y que actúan como posibles conclusiones en los razonamientos.

TÉRMINOS



Los términos compuestos no admiten espacios entre su nombre y “(” : por ejemplo “punto (X,Y)” produciría un error sintáctico.

- **Números:** notación usual, formato decimal o exponencial.
- **Átomos:**
 - cadenas de letras, dígitos o subrayado, **empezando por letra minúscula.**
 - o bien cadenas de caracteres entre comillas simples.
- **Variables:** cadenas de letras, dígitos o subrayado, **empezando por letra mayúscula o subrayado.** La variable `_` se denomina *variable anónima* y la variable `__`, junto con todas las que empiezan por subrayado seguido de letra mayúscula, son *variables semianónimas*. Las particularidades de estas variables se discuten más adelante.
- **Términos compuestos (estructuras):** constan de un nombre, que se denomina **functor** y se representa mediante un *átomo*, seguido, entre paréntesis, por una serie de **argumentos**, separados por comas, que son, a su vez, *términos* (constantes, variables o compuestos), por lo que los términos son *recursivos*. La **aridad** de un término compuesto es su número de argumentos.

PREDICADOS

- Tienen la misma sintaxis que los términos compuestos. Todo predicado tiene:
 - 1 Un **nombre o functor**, que se representa mediante un *átomo* y se puede sobrecargar (ver ejemplo a continuación).
 - 2 Cero o más **argumentos**, entre paréntesis y separados por comas, representados mediante *términos*.
- Establecen **propiedades** (si tienen menos de dos argumentos) o **relaciones** entre sus argumentos (cuando hay al menos dos).

Ejemplos

- llueve (predicado sin argumentos).
- animal(serpiente) (la serpiente es un animal).
- abuela(pepa, pepon) (pepa es abuela de pepon).
- abuela(pepa) (pepa es abuela). % sobrecarga de abuela

Convenios para describir predicados en libros y manuales

- `nombre_predicado/n`, siendo n la aridad (número de argumentos) del predicado.
- `nombre_predicado(<uso>Var_1, ..., <uso>Var_n)`, siendo `<uso>`:
 - `<uso> = +` : el argumento correspondiente debe estar, en el momento de usar el predicado, instanciado con un término *no variable* (parámetro de entrada).
 - `<uso> = -` : el argumento correspondiente *no* debe estar instanciado, es decir, debe ser una variable (parámetro de salida).
 - `<uso> = ?` : el argumento puede estar tanto instanciado como no instanciado (parámetro que se puede usar tanto para entrada como para salida).

PROGRAMAS

- Programa PROLOG = conjunto finito de **cláusulas de Horn positivas**, que pueden ser de dos clases: **hechos** o **reglas**.
- Hechos**. Los hechos se escriben en PROLOG mediante

$B.$ % acabado en punto

donde B es un *predicado*, y se corresponden con fórmulas lógicas de la Lógica de Predicados o Lógica de Primer Orden (LPO) de la forma

$$\forall X_1 \dots \forall X_p (B)$$

siendo X_1, \dots, X_p , $p \geq 0$, las variables que, en caso de que las haya, aparecen en B .

- **Reglas.** Las reglas se escriben en PROLOG como

```

B :-
    B1,
    ...,
    Bm.      % acabado en punto
  
```

siendo $B, B1, \dots, Bm$ *predicados*, y se corresponden con fórmulas lógicas de la LPO de la forma

$$\forall X_1 \dots \forall X_p \left((B1 \wedge \dots \wedge Bm) \rightarrow B \right)$$

donde $X_1, \dots, X_p, p \geq 0$, son las variables que aparecen en $B, B1, \dots, Bm$, en caso de que las haya.

B es la **cabeza** de la regla y $B1, \dots, Bm$ constituye su **cuerpo**.

Si $B \in \{B1, \dots, Bm\}$, se dice que la regla es **recursiva**.

Significado de hechos y reglas

- Significado de un hecho

$B.$

Para cualesquiera objetos X_1, \dots, X_p (variables que aparecen en B , si las hay), se cumple B .

- Significado de una regla

$B :-$

$B_1,$

$\dots,$

$B_m.$

Para cualesquiera objetos X_1, \dots, X_p (variables que aparecen en B, B_1, \dots, B_m , si las hay), *si* se cumplen B_1 y ... y B_m ($, = \wedge = y =$ conjunción) *entonces* ($\rightarrow =$ si...entonces = implicación) también se cumple B .

Ejemplo (Programa "Abuelas/os", hecho en LPO en PL1, 1/2)

```
% progenitor(?X,?Y)
% cierto si X es progenitor/a (padre o madre) de Y
% progenitor se describe mediante 3 HECHOS
```

```
    progenitor(pepa, pepito).
    progenitor(pepito, pepita).
    progenitor(pepito, pepon).
```

```
% abuelo(?X,?Z): cierto si X es abuela/o de Z
% abuelo/2 se describe mediante la siguiente REGLA
```

```
    abuelo(X,Z) :-
        progenitor(X,Y),
        progenitor(Y,Z).
```

Ejemplo (Programa "Abuelas/os" 2/2)

```
% abuelo(?X): cierto si X es abuela/o  
% abuelo/1 se describe mediante la siguiente REGLA  
  abuelo(X) :-  
    abuelo(X,_) .   % o abuelo(X,Y) .
```

Observación

- Definición de `abuelo/1`: para cualesquiera X e Y , si se cumple `abuelo(X, Y)` entonces se cumple `abuelo(X)`.
En LPO: $\forall X \forall Y (abuelo(X, Y) \rightarrow abuelo(X))$.
- La variable Y se puede reemplaza por la *variable anónima* `_` puesto que su valor no interesa (esto evita además el aviso del intérprete "singleton variable [Y]").

Ejemplo (Programa "Amistades/Enemistades")

```
% enemigo(?X,?Y): cierto si X es enemiga/o de Y
% enemigo se describe mediante 4 HECHOS
    enemigo(abel, cain).
    enemigo(cain, blas).
    enemigo(cain, dolores).
    enemigo(blas, abilio).

% amigo(?X,?Y): cierto si X es amiga/o de Y
% amigo se describe mediante 2 REGLAS
    amigo(abilio, X) :-
        amigo(abel, X).

    amigo(X,Y) :-
        enemigo(X,Z),
        enemigo(Z,Y).
```

Ejercicios (Programa "Amistades/Enemistades")

- 1 *Describa en Lógica de Primer Orden (LPO) y en lenguaje natural el conocimiento expresado por cada una de las dos reglas que definen el predicado `amigo`.*
- 2 *Escriba en LPO y en PROLOG el conocimiento expresado por las siguientes frases:*
 - *Los que son enemigos de Caín y de Abel también son enemigos de Abilio.*
 - *Dolores es amiga de cualquiera que sea enemigo de Abilio o de Blas.*
 - *Caín es enemigo de todos.*
- 3 *Describa en LPO y en lenguaje natural el conocimiento que describe el siguiente hecho:*
`amigo(abilio, X) .`
- 4 *Defina en PROLOG el predicado `tiene_enemigos(?X)`, cierto si `X` tiene algún enemigo.*

Ejemplo (Programa en lógica “pura” para sumar naturales)

- Un ejercicio clásico para ilustrar el uso y la potencia de PROLOG es la implementación de un programa para *sumar números naturales* con lógica “pura”, es decir, sin usar las facilidades aritméticas de PROLOG (que se estudian más adelante).
- La suma es una *función* matemática que recibe como entrada dos naturales y devuelve su suma. Para implementar una función en PROLOG, que solo dispone de predicados (relaciones), basta con definir un predicado con los mismos argumentos que la función pero añadiendo un argumento adicional para el resultado (colocado, por convenio, al final). En este caso:

```
suma (?X, ?Y, ?Z)
```

```
cierto si Z es la suma de X e Y
```

- Para trabajar en lógica “pura” se necesita un mecanismo para representar números naturales. Uno sencillo consiste en usar un símbolo, por ejemplo 0, para denotar el número natural cero, y otro, por ejemplo s, para construir el término compuesto s(X) representando el sucesor del número X, de forma que s(0) representa el número 1, s(s(0)) representa el 2, etc.
- Por otro lado, dado que PROLOG trabaja con recursividad, hay que partir de una definición *recursiva* de la suma:

Para cualquier $X \in \mathbb{N}$, $X + 0 = X$ (1)

Para cualesquiera $X, Y \in \mathbb{N}$, $X + (Y + 1) = (X + Y) + 1$ (2)

(1) establece que la suma de cualquier número con cero es él mismo, mientras que (2) asegura que si se conoce la suma de dos números, $X + Y$, entonces también se conoce la suma del primero con el sucesor del segundo, $X + (Y + 1)$, que no es otra cosa que el sucesor del valor $X + Y$, es decir, $(X + Y) + 1$.

- Lo anterior, utilizando el predicado y los símbolos definidos previamente, se expresa mediante las siguientes fórmulas:

$$\forall X \text{ suma}(X, 0, X) \quad (1)$$

$$\forall X \forall Y \forall Z \left(\text{suma}(X, Y, Z) \rightarrow \text{suma}(X, s(Y), s(Z)) \right) \quad (2)$$

En PROLOG:

```
% suma(?X,?Y,?Z): cierto si Z es la suma de X e Y

% (1) Caso base (expresado mediante un hecho)
    suma(X,0,X) .

% (2) Caso recursivo (expresado mediante una regla)
    suma(X,s(Y),s(Z)) :-
        suma(X,Y,Z) .
```

Ejercicios (Programación en lógica “pura”)

- 1 *Suponga que la primera cláusula de `suma` fuese `suma(0, X, X)` . en lugar de la dada. ¿Cómo sería el resto del programa?*
- 2 *Escriba sendos programas lógicos “puros” capaces de decidir si un número natural es par (impar). Es decir, implemente los predicados `par(?X)` e `impar(?X)` , ciertos si X es par (impar) sin usar aritmética y manejando los números naturales mediante la constante cero y la función sucesor $s(X)$ como se ha hecho para la suma.*
- 3 *Use el predicado `suma/3` para definir mediante lógica “pura” el predicado sobre números naturales `producto(?X, ?Y, ?Z)` , cierto si Z es el producto de X por Y . Parta de la siguiente definición recursiva del producto:*

Para cualquier $X \in \mathbb{N}$, $X \times 0 = 0$

Para cualesquiera $X, Y \in \mathbb{N}$, $X \times (Y + 1) = (X \times Y) + X$

CONSULTAS

Consultas para la activación de programas

Las **consultas** = **cláusulas de Horn negativas** = **cláusulas objetivo** = **cláusulas meta** tienen la siguiente sintaxis:

$$?- A1, \dots, An.$$

siendo $A1, \dots, An$ *predicados*, y se corresponden con fórmulas lógicas de la forma

$$\exists X_1 \dots \exists X_q (A1 \wedge \dots \wedge An)$$

siendo X_1, \dots, X_q , $q \geq 0$, las variables que, en caso de que las haya, aparecen en $A1, \dots, An$. Representan la pregunta *¿Es cierto que existen objetos X_1, \dots, X_q que cumplen $A1$ y ... y An ? En caso afirmativo, ¿qué objetos son?* ($, = \wedge = y =$ conjunción)

Ejemplos (Algunas consultas al programa “Abuelas/os”)

- `?- abuelo(pepa, pepito) . False`
- `?- abuelo(pepa, pepita) . True`
- `?- abuelo(pepa, N) . 2 soluciones: N=pepita y N=pepon`
- `?- abuelo(pepa, _) . True (variable anónima, no se computa)`
- `?- abuelo(A, pepon) . A=pepa`
- `?- abuelo(A, N) . 2 soluciones: [A=pepa,N=pepita] y [A=pepa,N=pepon]`
- `?- abuelo(X) . X=pepa`
- `?- progenitor(A, P), progenitor(P, pepita) .
A=pepa, P=pepito`
- `?- progenitor(X, pepito), progenitor(X, pepon) . False`
- `?- progenitor(X, pepon), progenitor(X, pepita) .
X=pepito`
- `?- abuelo(A, pepon), progenitor(P, pepon) .
A=pepa, P=pepito`

Observación (Uso de variables anónimas/semianónimas)

Ojo con el uso de la variable **anónima** (`_`) y las **semianónimas** (`__` o cualquiera que empiece por `_` seguido de una mayúscula):

- ?- `progenitor(pepa, P), progenitor(P, Z).` (v. normal)
2 soluciones: `[P=pepito, Z=pepita]` y `[P=pepito, Z=pepon]`
- ?- `progenitor(pepa, _P), progenitor(_P, Z).` (v. semi)
 La variable semianónima `_P` no se reporta pero sí tiene que unificar (“casar”) consigo misma:
2 soluciones: `Z=pepita` y `Z=pepon` (`_P = pepito` pero no se incluye)
- ?- `progenitor(pepa, _), progenitor(_, Z).` (v. anónima)
 La variable anónima ni reporta valor ni tiene por qué unificar (“casar”) en sus distintas apariciones, por lo que, dado que `progenitor(pepa, _)` se cumple para alguien, `Z` será cualquiera para el que se cumpla `progenitor(_, Z)`:
3 soluciones: `Z=pepito`, `Z=pepita` y `Z=pepon`

Ejemplos (Algunas consultas al programa “suma”)

- `?- suma(0, s(0), s(0)). True`
- `?- suma(s(0), s(0), X). X = s(s(0))`
- `?- suma(s(0), s(0), X), suma(X, s(s(0)), Z).
X = s(s(0)), Z = s(s(s(0)))`
- `?- suma(0, 0, Z), suma(0, s(s(0)), Z). False`
- `?- suma(s(s(0)), Y, s(0)). False`
- `?- suma(s(0), _, s(s(s(0)))) . True`
- `?- suma(s(0), Y, s(s(s(0)))) . Y = s(s(0))`

versatilidad de PROLOG: ¡“suma” también sirve para restar!

- `?- suma(X, Y, s(s(0))).`

3 soluciones: $[X = s(s(0)), Y = 0]$, $[X = s(0), Y = s(0)]$ y $[X = 0, Y = s(s(0))]$

versatilidad de PROLOG: ¡“suma” descompone en sumandos!

Ejercicios (Consultas al programa "Amistades/Enemistades")

1 *Expresar en LPO y en lenguaje natural qué se pretende averiguar con las siguientes consultas en PROLOG:*

- `?- amigo(abel, X).`
- `?- amigo(abel, _).`
- `?- enemigo(X, X).`
- `?- amigo(dolores, X), enemigo(X, dolores).`
- `?- amigo(X, _Y), amigo(X, _Z), enemigo(_Y, _Z).`

2 *Escriba en Lógica de Primer Orden y en PROLOG las siguientes preguntas:*

- *¿Tienen Abel y Caín algún amigo común? En caso afirmativo, ¿quién o quiénes?*
- *¿Es Caín amigo de alguno de los enemigos de Abel? En caso afirmativo, ¿quién o quiénes son?*
- *¿Tiene Abel enemigos?*
- *¿Qué parejas de personas existen tales que cada una de ellas es amiga de la otra?*

RESUMEN

	REGLAS	HECHOS
PROLOG	$B :- B_1, \dots, B_m.$	$B.$
LPO	$\forall X_1 \dots \forall X_p ((B_1 \wedge \dots \wedge B_m) \rightarrow B)$	$\forall X_1 \dots \forall X_p (B)$
Uso	en un programa	en un programa

	OBJETIVOS (METAS)
PROLOG	$?- A_1, \dots, A_n.$
LPO	$\exists X_1 \dots \exists X_q (A_1 \wedge \dots \wedge A_n)$
Uso	consulta

- B , B_i y A_j representan *predicados* y X_k *variables*.
- Hechos, reglas y consultas *terminan siempre con un punto*.
- Predicados: *empiezan por minúscula*.
- Variables: *empiezan por mayúscula o subrayado*.

Interpretación procedural de PROLOG

- 1 Predicados = *procedimientos*, con argumentos = *parámetros*.
- 2 Reglas = *definiciones de procedimientos*:

Una regla

$$B \text{ :- } B1, B2, \dots, Bm.$$

se puede interpretar de la siguiente forma: computar el procedimiento B equivale a computar en primer lugar el procedimiento B1, a continuación el procedimiento B2, etc hasta Bm.

- 3 Consultas = *llamadas a procedimientos*:

Una consulta

$$?- A1, A2, \dots, An.$$

se puede interpretar como una llamada al procedimiento A1, seguida de una llamada al procedimiento A2, etc hasta An.

Ejercicios (Sintaxis de PROLOG)

- *Ejercicio nº 1 de la **Práctica de PROLOG nº 1** (puede consultar las instrucciones generales para la realización de las prácticas **aquí**).*
- *Ejercicio nº 2 de la **Práctica de PROLOG nº 1**.*

*Las soluciones propuestas para ambos ejercicios, comentadas, están disponibles en **Práctica de PROLOG nº 1 con soluciones**.*

BIBLIOGRAFÍA

- L. Sterling and E. Shapiro. *The Art of Prolog*. The MIT Press, Cambridge, Mass., second edition, 1994.
- W.F. Clocksin and C.S. Mellish. *Programming in Prolog*. Springer-Verlag, Berlin, fifth edition, 2003.
- I. Bratko. *Prolog Programming for Artificial Intelligence*. Addison-Wesley, Reading, Massachusetts, third edition, 2001.
- J. Lloyd. *Foundations of Logic Programming*, (Second Edition). Springer-Verlag, 1987.
- R. O'Keefe. *The Craft of Prolog*. The MIT Press, Cambridge, MA, 1990.
- U. Nilsson and J. Maluszynski. **Logic, Programming and Prolog**. John Wiley & Sons Ltd, 1996.
- **SWI-Prolog**, entorno de programación en Prolog de dominio público.
- **comp.lang.prolog. Faq**

© 2022 Ana Pradera Gómez

Algunos derechos reservados

Este documento se distribuye bajo la licencia

“Atribución-CompartirIgual 4.0 Internacional” de Creative Commons,
disponible en

<https://creativecommons.org/licenses/by-sa/4.0/deed.es>